

# Translating Chalice into SIL

## Report

Christian Klauser  
klauserc@student.ethz.ch

November 2, 2012

- Use either `acc(x.f, write)` (SIL toString) or `acc(x.f, full)` (SIL API), not both!
- Don't use abbreviated read fraction syntax `rd(x.f)`, ever.

## 1. Introduction

### 1.1 Semper Project

## 2. Background

### 2.1 Chalice

Chalice is a research programming language with the goal of helping programmers detect bugs in their concurrent programs. As with most languages aimed at automatic static verification (e.g. Spec#), the programmer provides annotations that specify how they intend the program to behave. These annotations appear in the form of monitor invariants, loop invariants and method pre- and postconditions. A verification tool can take such a Chalice program and check statically that it never violates any of the conditions established by the programmer.

The original implementation of the automatic static program verifier for Chalice generates a program in the intermediate verification language Boogie [BDJ<sup>+</sup>06]. A second tool, conveniently also called Boogie, takes this intermediate code and generates verification conditions to be solved by an SMT solver, such as Z3 [dMB08].

Listing 1 demonstrates how we can implement integer division and have the verifier ensure that our implementation is correct. Our solution repeatedly subtracts the denominator `b` until the rest `r` becomes smaller than `b`. Because this exact algorithm only works for positive numerators and denominators, the method **requires** that the numerator `a` is not negative and that the denominator is strictly positive.

Similarly, we specify what the method is supposed to do: the **ensures** clause tells the verifier that, when our method is ready to return, the resulting quotient `c` must be the largest

Listing 1: Loop invariants, pre- and post conditions in a Chalice program

```
class Program {
  method intDiv(a : int, b : int) returns (c : int)
    requires 0 <= a && 0 < b;
    ensures c*b <= a && a < (c+1)*b;
  {
    c := 0;
    var r : int := a;
    while(b <= r)
      invariant 0 <= r && r == (a - c*b)
    {
      r := r - b;
      c := c + 1;
    }
  }
}
```

Listing 2: Chalice example of object creation and (write) accessibility predicates.

```
class Cell { var f : int }
class Program {
  method clone(c : Cell) returns (d : Cell)
    requires c != null && acc(c.f)
    ensures acc(c.f)
    ensures d != null && acc(d.f) && d.f == c.f
  {
    d := new Cell;
    d.f := c.f;
  }
}
```

integer for which  $c \cdot b \leq a$  still holds. If the verifier cannot show that this postcondition holds for all invocations of this method that satisfy the precondition, it will reject the program.

The final bit of annotation in this example is the **invariant** on the **while** loop. A loop invariant is a predicate that needs to hold immediately before the loop is entered and after every iteration, including the last one where the loop condition is already false. This annotation helps the verifier understand the effects of the loop without knowing how many iterations of the loop would happen at runtime.

### 2.1.1 Permissions

What sets Chalice apart from other languages for program verification is its handling of concurrent access to heap locations. Whenever a thread wants to read from or write to a heap location it requires read or write permissions to that location, respectively. These permissions only exist for verification and would be erased by compilers for Chalice.

Listing 3: Calling `Program::clone` (extension of Listing 2)

```

1 class Program {
2   //...
3   method main()
4   {
5     var c : Cell := new Cell;
6     c.f := 5;
7     var d : Cell;
8     call d := clone(c);
9     assert d.f == 5; // will fail, c.f might have changed
10  }
11 }

```

As an under-approximation of the set of permissions a thread would have at runtime, Chalice tracks permissions for each method invocation (stack frame, activation record). That way, the verifier can verify method bodies in complete isolation of one another. The programmer thus has to specify which heap locations need to be accessible for each method.

In Listing 2, we use *accessibility predicates* of the form `acc(receiver.field)` in the method’s pre- and postcondition. `acc(c.f)` in the precondition allows us to refer to `c.f` in the method body. The accessibility predicates in the postcondition, on the other hand, represent permissions that the method will have to “return” to its caller upon completion. Conceptually, the caller passes the permission requested by the callee’s precondition on to the callee. Similarly, the caller receives the permissions mentioned in the callee’s postcondition when the call returns. As a consequence of verifying each method in isolation, it doesn’t matter whether a method is called on the same thread or on a thread of its own (with the caller waiting for the callee’s computation to finish). The necessary permissions need to be transferred in both scenarios.

Listing 3 demonstrates how our `clone` method could be used. Unfortunately, the assertion on line 9 will fail, as the verifier has to assume that `clone` might have changed the value stored in `c.f`. In Chalice, whenever a method gives away all permissions to a memory location (so that it doesn’t even have read-access), it must assume that that location has been changed, the next time it gets to read said location. While we might change augment the postcondition of `clone` with the requirement that `c.f == old(c.f)` (the value of `c.f` at method return must be the same as it was on method entry), there is a much more elegant solution to this problem: *read-only permissions*.

### 2.1.2 Percentage Permissions

When Chalice was originally created, the programmer could specify read-only permissions as *integer percentages* of the full (write) permission. `acc(x.f, 100)` is the same as `acc(x.f)`, i.e. grant read and write access, whereas any other strictly positive percentage `acc(x.f, n)` for only grants read access to the heap location `x.f` ( $n \in \mathbb{N}, 0 < n < 100$ ). The verifier keeps track of the exact amount of permission a method holds to each heap location, so that write-access is restored when a method manages to get 100% of the permission back together, after having handed out parts of it to other methods or threads.

While percentage permissions are very easy to understand, they have the serious draw-

back that the number of percentage points of permission a method receives to a certain location, essentially determine the maximum number of threads with (shared) read access that method could spawn. That is a violation of the procedural abstraction that methods are intended to provide.

### 2.1.3 Counting Permissions

Another drawback of percentage permissions is, that it is difficult to deal with a dynamic number of threads to distribute read access over. As a solution to that problem, Chalice also introduced “*counting permissions*” that are not limited to just 100 “pieces” of permission. Accessibility predicates using counting permissions are written as  $\text{acc}(x.f, \text{rd}(1))$  and denote an arbitrarily small but still positive (non-zero) amount of permission  $\varepsilon$ . Permission amounts equal to multiples of  $\varepsilon$  can be written as  $\text{acc}(x.f, \text{rd}(n))$ , but any finite number of epsilon permissions are defined to be still smaller than 1% of permission. This also means that a method that holds at least 1% of permission, can always call a method that only requires  $n \cdot \varepsilon$  of permission.

Unfortunately, counting permissions (often also referred to as “*epsilon permissions*”) still cause method specifications to leak implementation details. An epsilon permission cannot be split up further, thus a method that acquires, say,  $2\varepsilon$  of permission to a heap location cannot spawn more than two threads with read access to that heap location.

### 2.1.4 Fractional (Read) Permissions

In order to regain procedural abstraction [HLMS11] added an entirely new kind of permission to Chalice: the fractional read permission, based on [Boy03]. The idea is to allow for “rational” fractions of permission because, unlike epsilon or percentage permissions, those could always be divided further. Composability can still be an issue, even with rational permissions. A method that requires  $\frac{1}{107}$  of permission could still not be called from a method that only has  $\frac{1}{137}$ , even though the fractions passed around the entire system could almost always be re-scaled to make that call possible. Thus, instead of forcing the programmer to choose a fixed amount of permission ahead of time, all accessibility predicates involving fractional permissions are kept *abstract*.

The programmer writes  $\text{acc}(x.f, \text{rd})$  to denote an abstract (read-only) accessibility predicate to the heap location  $x.f$ . The amount of permission denoted by  $\text{rd}$  is not fixed. When used in a method specification, the  $\text{rd}$  can represent a different amount of permission for each method invocation.

To make fractional permissions actually useful, Chalice applies certain constraints to the amount of permission involved in  $\text{acc}(x.f, \text{rd})$ . Firstly, fractional read permissions always represent a fraction of the caller’s permission. When a caller gives away a fractional read permission to a heap location, it will always retain read-access to that location. That way, the caller can be sure that the contents of the memory location don’t change. Secondly, a common idiom in Chalice is to have methods that return the exact same permissions they acquired in the precondition back to the caller via the postcondition. When a method requires  $\text{acc}(x.f, \text{rd})$  and then ensures  $\text{acc}(x.f, \text{rd})$ , we would want these two amounts of permission to be the same. That way, a caller that started out with write access to  $x.f$  gets back the exact amount of permission it gave to our method.

Listing 4: Corrected example using abstract read permissions

```

1 class Cell { var f : int }
2 class Program {
3   method clone(c : Cell) returns (d : Cell)
4     requires c != null && acc(c.f,rd)
5     ensures acc(c.f,rd)
6     ensures d != null && acc(d.f) && d.f == c.f
7   {
8     d := new Cell;
9     d.f := c.f;
10  }
11
12  method main()
13  {
14    var c : Cell := new Cell;
15    c.f := 5;
16    var d : Cell;
17    call d := clone(c);
18    assert d.f == 5; // will now succeed
19    c.f := 7; // we still have write access
20  }
21 }

```

Chalice restricts read fractions in method specifications even further: all fractional read permissions in a method contract, even to different heap locations, refer to the same amount of permission (but that amount can still differ between method invocations). This restriction accounts for the limited information about aliasing available statically and also makes the implementation of fractional read permissions more straightforward.

Listing 4 shows the corrected version of our example above (Listings 2 and 3) using (abstract) read permissions (`acc(c.f,rd)` in lines 4 and 5). Note that we don't need to tell the verifier that `c.f` won't change separately, because it uses permissions to determine what locations can be modified by the method call to `clone`.

### 2.1.5 Fork-Join

As a language devoted to encoding concurrent programs, Chalice has a built-in mechanism for creating new threads and waiting for threads to complete in the familiar *fork-join* model. Replacing the `call` keyword in a (synchronous) method call with `fork` causes that method to be executed in a newly spawned thread. As with a synchronous method call, the caller must satisfy the callee's precondition and will give all permissions mentioned in that precondition.

```

fork tok := x.m(argument1, argument2, ..., argumentn);
// do something else
join result1, result2, ..., resultn := tok;

```

Listing 5: Alternative definition of `Cell` using functions.

```
class Cell {  
  var f : int  
  function equals(o : Cell) : bool  
    requires acc(f,rd)  
    requires o != null ==> acc(o.f,rd)  
    { o != null && f == o.f }  
}
```

While just forking off threads might work for some scenarios, most of the time the caller will want to collect the results computed by its worker threads at some point. To that end, the `fork` statement will return a *token* that the programmer can use to have the calling method wait for the thread associated with the token to complete. The permissions mentioned in the postcondition of the method used to spawn off the worker thread will also be transferred back to the caller at that point.

### 2.1.6 Information Hiding through functions and predicates

A major shortcoming of pre- and postconditions as presented so far, is that they often “leak” implementation details. One example of this happening is the `clone` method from listing 4. It ensures that the values from the old object are copied over to the newly created object, but in the process tells the caller that there is exactly one field, called `f` on those objects. Should the definition of class `Cell` ever change, sifting through the entire program and updating specifications is going to be in order. What the programmer wanted to say is, that the two objects are “*equal*”.

**Functions** help cut down code repetition and put an abstraction layer between the implementation of a method and its clients. Listing 5 presents an alternative definition of `Cell` that exposes the equality testing function `equals`. Below is a corresponding signature for the method `clone` that uses this function. If we were to add a new field to `Cell` now, callers of `clone` would no longer see a change in the method’s signature.

```
method clone(c : Cell) returns (d : Cell)  
  requires c != null && acc(c.f,rd)  
  ensures acc(c.f,rd)  
  ensures d != null && acc(d.f) && c.equals(d)
```

Notice how the `equals` function does *not* have a postcondition that describes the function’s result or “returns” permissions back to the caller. This is because functions are little more than abbreviations of common expressions. In order to be used in pre- and postconditions, they are forbidden from changing any state, which is why the programmer doesn’t have explicitly return permissions to the function’s caller. This happens automatically.

**Predicates**, on the other hand, are a way to abstract over not just values but also over accessibility. Additionally, unlike functions, they are treated as abstract entities unless the programmer explicitly “unfolds” them to apply their definition. When a method requires a predicate in its precondition, it will not automatically get the permissions (and other assertions) “contained” in the predicate because at that point, the predicate acts like a black box. The method can pass the predicate to other methods or threads and it behaves

Listing 6: Using the predicate `valid` to hide the representation of Indentation

```
class Indentation {
  var count : int;

  predicate valid
  { acc(count) && 0 <= count }

  function getCount() : int
  requires valid;
  { unfolding valid in count }

  method increase(amount : int)
  requires valid && 0 <= amount;
  ensures valid;
  ensures old(getCount()) + amount == getCount();
  {
    unfold valid;
    count := count + amount;
    fold valid;
  }
}
```

much like a permission to a memory location: it cannot be duplicated and once given away, it's gone.

Given a predicate, the programmer can use the `unfold` statement to “trade” the predicate for its definition. The current thread will receive all permissions “contained” in the predicate and gets to assume any other assertions associated with the predicate. After the programmer is done operating on the predicate’s contents, they can use `fold` to “trade” access permissions in exchange for the predicate.

Listing 6 additionally demonstrates the `unfolding` expression syntax used to temporarily getting access to the contents of a predicate.

### 2.1.7 Monitors (locks)

Using just fork-join, it is impossible for threads to communicate with one another. They can only produce a result and all of their memory writes only become visible when they return the exclusive write permissions back to their caller. To handle more realistic scenarios, such as concurrent access to a shared queue, Chalice comes with *monitors* that allow for exclusive locking of a shared resource. For each class, the programmer can define a *monitor invariant* that represents the “resources” that the monitor is supposed to manage access to. As with predicates, this definition can consist of both accessibility predicates and ordinary boolean assertions.

Initially, objects are not available for locking via the monitor mechanism. When the programmer *shares* an object with other threads using the `share` statement, the access permissions associated with the invariant get stored in the monitor (similar to `fold` for predicates). Threads that subsequently `acquire` the lock on this *shared* object will receive the

Listing 7: Example of the life-cycle an object can go through in Chalice

```
class C {
  var f : int;

  invariant acc(f);

  method main(){
    var c : C := new C;
    c.f := 5;
    share c;
    acquire c; c.f := 7; release c;
    // cannot access c.f here
    acquire c; c.f := 6; unshare c;
    assert c.f == 6;
  }
}
```

contents of the monitor (similar to **unfold**). The object is now *locked* and can be made available to other threads via the **release** statement (similar to **fold**, again). The programmer can also revert the conversion to a *shared* object by using the **unshare** statement (similar to **unfold**, again). Listing 7 demonstrates these statements with a single thread.

As with monitors in Java and C#, in order to guarantee mutual exclusion, threads that reach an **acquire** statement are blocked until the monitor can grant them the exclusive lock. With such a simple blocking mechanism comes the risk of deadlocks (thread 1 waiting for monitor *b*, currently held by thread 2, which is waiting for monitor *a*, currently held by thread 1).

To solve this problem, the Chalice verifier makes sure that locks are acquired in a fixed order. The programmer can assign a *locking level* to a monitor, ensuring that the lock on that monitor can only be acquired when that locking level is *higher* than the locking level of all other locks held by the current thread. Whether one locking level is higher than another, is determined by a strict partial order that we denote as  $<<$ . The **share** statement seen above optionally accepts clauses of the form **between** ... and ..., **above** ... or **below** ... to constrain the *lock level* at which the monitor is installed. If such a clause is missing, Chalice chooses `above waitlevel`, which means that the lock level is higher than the highest lock level of all locks currently held by the thread (we refer to this maximum as a thread's *wait level*).

In listing 8, we create two objects *a* and *b* and share them. The lock level of *a* defaults to `above waitlevel` and the programmer explicitly declare the lock level of *b* to be `above a`. This means that if a thread plans to lock both *a* and *b*, it will have to first lock on *a* then *b*. Should the programmer try to lock objects in the opposite order, on **acquire** *a* the thread's wait level would already be at the lock level of *b*, which is `above a`'s.

Lock levels are implemented via a special field called `mu` of type `Mu` (the type of lock levels), available on every object. The `mu` field is assigned during **share** and **unshare** operations and needs to be read-able for acquiring the lock.



Listing 8: Example of deadlock-prevention

```
class C {
  var f : int;
  invariant acc(f);

  method main() {
    var a := new C;
    share a;
    var b := new C;
    share b above a;

    acquire a; acquire b;
    release b; release a;

    acquire b;
    acquire a; // illegal
  }
}
```

### 2.1.8 Details on the Boogie-based Chalice verifier

Is “Inhale, exhale and havoc” a better title for this section?

In order to verify Chalice programs, the Boogie-based verifier permission transfer is modelled by two operations: **inhale** and **exhale**. They are essentially the same as **assume** and **assert** but in addition to providing and checking facts, they also model the transfer of permissions. The argument of an **exhale** operation is an expression that can contain both traditional (boolean) assertions as well as accessibility predicates. Conceptually, **exhale**  $e$  represents the transfer of  $e$  to another thread. Because verification of Chalice methods is modular, we don’t specify or even care about which thread will “receive”  $e$ . For each **exhale** operation, the verifier will check (assert) the boolean predicates and remove permissions mentioned in  $e$  from the current thread’s set of permissions (usually referred to as the thread’s “*permission mask*”). The **inhale**  $e$  operation works the opposite way. Access permissions mentioned in  $e$  are added to the thread’s permission mask and boolean predicates get assumed.

Where is the best place for explaining the idea that as soon as you give away all permissions to a location, anything can happen to that location, from your perspective?

## 2.2 Semper Intermediate Language (SIL)

The Semper Intermediate Language is a verification language aimed at the verification of concurrent programs using a methodology based on Chalice. As its name suggests, SIL is the intermediate language to be used by the various tools that are part of the semper project.

Much of SIL’s design is oriented around Chalice’s core elements: methods, permissions and accessibility predicates. This also means that SIL programs are encoded on a much

higher level of abstraction than the same programs in less focused verification languages, such as Boogie. As an example: the Boogie-based verifier for Chalice needs to represent permissions as a pair of integers (the number of epsilons and the percentage) whereas in SIL there is a dedicated and built-in data type and associated value constructor functions for permissions.

In this section, we will give an overview of the syntactical structure of SIL programs, diving into more detail where the design of SIL deviates significantly from Chalice. At this time SIL is mostly intended as an “exchange format” and thus has no fixed semantics associated with it. Also, SIL doesn’t currently have a serialised/text form and SIL programs only exist as syntax trees in memory. As a result we use our own ad-hoc textual representation for SIL program snippets in this report.

### 2.2.1 SIL Program Structure

Each SIL program has a name ( $\langle program-id \rangle$ ) and comes with a number of *domain*, *field*, *function*, *predicate* and *method definitions*. While SIL is certainly aimed at the verification of object oriented programs, it isn’t actually necessary to distinguish between the types of references to objects created from different classes. As a direct result, fields, functions, predicates and methods are not “contained” in any form of class definition.

```

 $\langle Program \rangle ::= \text{'program' } \langle program-id \rangle$ 
    { $\langle Domain \rangle$ }
    { $\langle Field \rangle$ }
    { $\langle Function \rangle$ }
    { $\langle Predicate \rangle$ }
    { $\langle Method \rangle$ }

```

Field and predicate definitions, apart from the fact that they are not tied to a nominal class, are fairly straightforward. Fields consist of a name and a data type and predicates consist of a name and an expression. As with Chalice, this predicate expression can contain both accessibility predicates and ordinary boolean predicates. Field and predicate names must each be unique within a SIL program.

```

 $\langle Field \rangle ::= \text{'field' } \langle field-id \rangle \text{' : ' } \langle DataType \rangle$ 

```

```

 $\langle Predicate \rangle ::= \text{'predicate' } \langle pred-id \rangle \text{' = ' } \langle Expr \rangle$ 

```

Functions, again, are similar to their Chalice counterparts. They consist of a name, a parameter list, a result type, some preconditions and an implementation. Note how an  $\langle Expr \rangle$  is expected for the preconditions and a  $\langle Term \rangle$  for the function’s body. That is SIL distinguishing syntactically between assertions/formulae ( $\langle Expr \rangle$ ) and expressions that represent a value ( $\langle Term \rangle$ ).

```

 $\langle Function \rangle ::= \text{'function' } \langle id \rangle ( \{ \langle Param \rangle , \dots \} ) : \langle DataType \rangle$ 
     $\langle Contract \rangle \text{' = ' } \langle Term \rangle$ 

```

```

 $\langle Param \rangle ::= \langle id \rangle : \langle DataType \rangle$ 

```

```

 $\langle Contract \rangle ::= \{ \text{'requires' } \langle Expr \rangle \} \{ \text{'ensures' } \langle Expr \rangle \}$ 

```

Methods in SIL have a name (unique among all methods in the program), input and output parameters and a set of pre- and postconditions. Every SIL method always has a parameter called `this` of type `ref` in the first position, which represents the `this` pointer in object oriented languages. Having the `this` pointer as an ordinary parameter makes tools that

consume SIL programs a bit simpler. Each method can have multiple implementations that must all share the exact same parameters, pre- and postconditions. For source languages with virtual methods, the to-SIL-translator would create a method for each “method slot” (vtable slot) and add an implementation for each concrete implementation encountered in the program.

$\langle \text{Method} \rangle ::= \text{'method'} \langle \text{method-id} \rangle ( \{ \langle \text{Param} \rangle , \dots \} ) : ( \{ \langle \text{Result} \rangle , \dots \} )$   
 $\langle \text{Contract} \rangle \{ \langle \text{Impl} \rangle \}$

$\langle \text{Result} \rangle ::= \langle \text{Param} \rangle$

$\langle \text{Impl} \rangle ::= \text{'implementation'} \langle \text{method-id} \rangle \langle \text{Cfg} \rangle$

Method bodies in SIL are represented as a control-flow graph. This is mostly because SIL is intended as a format for exchanging programs between the tools that make up Semper as opposed to an actual computer languages used by humans. Whether an eventual textual representation would retain this form, is not clear at this point.

Unusual about SIL’s control-flow graph is that loops are loops are not flattened into basic blocks but retained as a sort of composite block. A loop block consists of the loop condition, an invariant and a nested control-flow graph for the loop’s body.

$\langle \text{Cfg} \rangle ::= \text{'{' } \{ \langle \text{VarDecl} \rangle \} \{ \langle \text{Block} \rangle \} \text{'}'}$

$\langle \text{VarDecl} \rangle ::= \text{'var'} \langle \text{var-id} \rangle : \langle \text{DataType} \rangle$

$\langle \text{Block} \rangle ::= \langle \text{BasicBlock} \rangle$   
 $\quad | \quad \langle \text{LoopBlock} \rangle$

$\langle \text{LoopBlock} \rangle ::= \text{'while'} \langle \text{PExpr} \rangle [ \text{'invariant'} \text{ Expr } ] \text{'do'} \langle \text{Cfg} \rangle$

$\langle \text{BasicBlock} \rangle ::= \langle \text{label} \rangle : \text{'{' } \{ \langle \text{Stmt} \rangle \} \langle \text{ControlFlow} \rangle \text{'}'}$

$\langle \text{ControlFlow} \rangle ::= \text{'goto'} \langle \text{label} \rangle$   
 $\quad | \quad \text{'halt'}$   
 $\quad | \quad \text{'if'} \langle \text{PExpr} \rangle \text{'then goto'} \langle \text{label} \rangle \text{'else goto'} \langle \text{label} \rangle$

At the end of every block there is a single control-flow statement that indicates how control is transferred to other blocks.

## 2.2.2 SIL Statements

$\langle \text{Stmt} \rangle ::= \langle \text{var-id} \rangle \text{' := ' } \langle \text{PTerm} \rangle$   
 $\quad | \quad \langle \text{var-id} \rangle . \langle \text{field-id} \rangle \text{' := ' } \langle \text{PTerm} \rangle$   
 $\quad | \quad \langle \text{var-id} \rangle \text{' := new' } \langle \text{DataType} \rangle$   
 $\quad | \quad \vdots$

$\langle \text{Stmt} \rangle ::= \vdots$   
 $\quad | \quad ( \{ \langle \text{var-id} \rangle , \dots \} ) \text{' := ' } \langle \text{PTerm} \rangle . \langle \text{method-id} \rangle ( \{ \langle \text{PTerm} \rangle , \dots \} )$   
 $\quad | \quad \vdots$

$\langle \text{Stmt} \rangle ::= \vdots$   
 $\quad | \quad \text{'inhale'} \langle \text{Expr} \rangle$   
 $\quad | \quad \text{'exhale'} \langle \text{Expr} \rangle$   
 $\quad | \quad \vdots$

$\langle \text{Stmt} \rangle ::= \vdots$   
 $\quad | \quad \text{'fold'} \langle \text{Term} \rangle . \langle \text{pred-id} \rangle \text{'by'} \langle \text{Term} \rangle$   
 $\quad | \quad \text{'unfold'} \langle \text{Term} \rangle . \langle \text{pred-id} \rangle$

### 2.2.3 SIL Expressions and Terms

```

⟨Expr⟩ ::= 'acc' (⟨Location⟩, ⟨Term⟩)
| 'old' (⟨Expr⟩)
| 'unfolding' ⟨Term⟩.⟨pred-id⟩ 'by' ⟨Term⟩ 'in' ⟨Expr⟩
| ⟨Term⟩ == ⟨Term⟩
| ⟨unary-op⟩ ⟨Expr⟩
| ⟨binary-op⟩ ⟨Expr⟩
| ⟨dom-pred-id⟩({ ⟨Term⟩, ... })
| ∀ ⟨logical-var-id⟩ : ⟨DataType⟩ :: (⟨Expr⟩)
| ∃ ⟨logical-var-id⟩ : ⟨DataType⟩ :: (⟨Expr⟩)

⟨Location⟩ ::= ⟨Term⟩.⟨field-id⟩
| ⟨Term⟩.⟨pred-id⟩

⟨Term⟩ ::= 'if' ⟨Term⟩ 'then' ⟨Term⟩ 'else' ⟨Term⟩
| ⟨var-id⟩
| ⟨logical-var-id⟩
| 'old' (⟨Term⟩)
| ⟨func-id⟩({ ⟨Term⟩, ... })
| ⟨dom-func-id⟩({ ⟨Term⟩, ... })
| 'unfolding' ⟨Term⟩.⟨pred-id⟩ 'by' ⟨Term⟩ 'in' ⟨Term⟩
| (⟨Term⟩) : ⟨DataType⟩
| ⟨Term⟩.⟨field-id⟩
| 'perm' (⟨Location⟩)
| 'write'
| '0'
| 'E'
| ⟨integer-literal⟩

```

We simplified the presentation of the term and expression grammar for this section and attached the full rules in appendix [A](#).

### 2.2.4 SIL Domains and Types

A data type in SIL is either 'ref', the type of all object references, a domain type or a type variable (only for data types in domain templates). Object references in SIL are treated as potentially having all fields in the SIL program. In practice, only the fields that a method-/function has access to, are relevant. For statically types programming languages, it's the responsibility of the to-SIL-translator to make sure that input programs are type error free.

```

⟨DataType⟩ ::= ⟨var-type⟩
| ⟨dom-type⟩
| 'ref'

```

In addition to the built-in value domains for integers, booleans and permissions, SIL allows its users to define their own value domains, with (uninterpreted) constructor functions, predicates over values of that domain and their axioms. Domain definitions can come with type parameters, making them templates for concrete domains (similar to C# generics).

```

⟨Domain⟩ ::= 'domain' ⟨dom-id⟩ [ ⟨DomainParameters⟩ ] '{' ⟨DomainDef⟩ '}'
⟨DomainDef⟩ ::= { ⟨DomainFunction⟩ } { ⟨DomainPredicate⟩ } { ⟨DomainAxiom⟩ }
⟨DomainFunction⟩ ::= 'function' ⟨dom-func-id⟩ ( { ⟨DataType⟩ , ... } ) : ⟨DataType⟩
⟨DomainPredicate⟩ ::= 'predicate' ⟨dom-pred-id⟩ ( { ⟨DataType⟩ , ... } )
⟨DomainAxiom⟩ ::= 'axiom' ⟨id⟩ '=' ⟨DExpr⟩
⟨DomainParameters⟩ ::= '[' { ⟨DataType⟩ , ... } ']'

```

### 3. Translation of Chalice

High-level overview + including focus areas

#### 3.1 Fractional Read Permissions

To SIL, permission amounts are just another data type. The SIL prelude only defines a set of constructors (such as no permission, full permission) and some operators and predicates (such as permission addition, subtraction, equality, comparison). In particular, it does not specify how permissions are represented. This aligns well with the abstract nature in which fractional permissions are written by the programmer. As with previous verification backends for Chalice, concrete permission amounts associated with fractional read permissions ( $\text{acc}(x.f, rd)$ ) are never chosen but only constrained. This also means that two textual occurrences of  $\text{acc}(x.f, rd)$  in different parts generally do not represent the same amount of permission.

Not choosing a fixed permission amount for abstract read permissions makes them very flexible. As long as a thread holds any positive amount of permission to a location, we know that we can give away a smaller fraction to a second thread and thereby enable both threads to read that location. Unfortunately, that amount of flexibility would also make fractional read permissions very hard to use, since every mention of a read permission could theoretically refer to a different amount of permission. Chalice, therefore, imposes additional constraints on fractional permissions involved in method contracts, predicates, and monitors. In the following sections we will describe how Chalice2SIL handles each of these situations.

##### 3.1.1 Methods and fractional permissions

In Chalice programs, a very common pattern is that a method “borrows” permissions to a set of locations, performs its work and then returns the same amount of permission to the method’s caller. In order to readily support this scenario, the original implementation of fractional permissions in Chalice constrains the various fractions mentioned in a method’s pre- and postcondition to a value that is chosen once per call site.

For verifying the callee in listing 9, the Boogie-based implementation introduces a fresh variable permission variable  $k_m$ , constrains it to be a read-permission ( $0 < k_m < \text{full}$ ) and uses it in pre- and postconditions whenever it encounters the abstract permission amount  $rd$ .

Listing 9: A call that uses and preserves fractional read permissions.

```
class Actor {
  method main(a : int) returns (r : Register)
    ensures r != null
    ensures acc(r.val)
    ensures t.val == a
  {
    r := new Register;
    r.val := 5;
    call act(r);
    r.val := a; //should still have write access here
  }

  method act(r : Register)
    requires r != null
    requires acc(r.val,rd)
    ensures acc(r.val,rd)
  { /* ... */ }
}

class Register {
  var val : int;
}
```

Listing 10: Handling of fractional read permissions by the Boogie-based Chalice verifier.

```
procedure act(r : Register)
{
  var k_m;
  assume (0 < k_m) && (k_m < Permission$FullFraction);
  // inhale (precondition), using k_m for rd
  ...
  // exhale (postcondition), using k_m for rd
}
```

Listing 11: Handling of fractional read permissions by the Chalice2SIL translator

```
method Actor::act(r : Register, k_m : Permission)
  requires 0 < k_m && k_m < write
  requires r != null
  requires acc(r.val, k_m)
  ensures acc(r.val, k_m)
{ ... }
```

Notice how the Boogie-based encoding of Chalice in listing 10 does not make use of the pre- and postcondition mechanism provided by Boogie. This is primarily because Boogie does not have a concept of inhaling and exhaling of permissions. Not so with SIL, which features pre- and postconditions that are aware of access predicates. Conceptually, when you “call” a method in SIL, the precondition is properly exhaled and the postcondition inhaled afterwards.

However, using SIL preconditions also means that we can’t just make up a new variable  $k_m$ , instead it becomes a “ghost” parameter and introduces an additional precondition. This makes a lot of sense, since the value  $k_m$  is always specific to one call of a method.

In the actual Boogie-based encoding, the upper bound on  $k_m$  is even lower to give the programmer more flexibility. Currently,  $k_m$  is assumed to be smaller than a thousandth of 1%. This allows the programmer to for instance specify a method that requires `acc(x.f, rd)` twice, effectively demanding at least  $2*k_m$  permission to `x.f`. The exact ratio was chosen arbitrarily and could always be lowered, but has so far worked well for most examples.

### 3.1.2 Method calls with fractional permissions

Without fractional permissions, synchronously calling a method in SIL is as simple as using the built-in call statement:

```
call () := Actor::act(r)
```

SIL takes care of asserting the precondition, exhaling the associated permissions, havocing the necessary heap locations, inhaling the permissions mentioned by the postcondition and finally assuming said postcondition. Adding support for fractional read permissions now only means providing a call-site specific value  $k$ , right?

Unfortunately, this where the high-level nature of SIL becomes an obstruction. For each method call-site, we want to introduce a fresh variable  $k_c$  that represents the fractional permission amount of permission selected for that particular call. Then, we want to constrain it to be smaller than the amount of permissions we hold to each of the locations mentioned with abstract read permissions (`rd`). For the simple preconditions above, this is easy to accomplish:

```
var k_c : Permission;
assume k_c < perm(r.val);
call () := Actor::act(r, k_c);
```

The term  $\text{perm}(r.\text{val})$  is a native SIL term that represents the amount permission the current thread holds to a particular location. Sadly, this simple scheme breaks down when we have to deal with multiple instances of access predicates to the same location.

Chalice dictates that

```
exhale acc(x.f, rd) && acc(x.f, rd)
```

is to be treated as

```
exhale acc(x.f, rd)
exhale acc(x.f, rd)
```

Both exhale statements cause  $k_c$  to be constrained to the amount of permission held to  $x.f$ . Since exhale has the “side-effect” of giving away the mentioned permissions, this  $k_c$  will be constrained further by the second exhale statement.

Additionally, access predicates can be guarded by implications. In that case, the Boogie-based Chalice implementation translates

```
exhale P ==> acc(x.f, rd)
```

as

```
if(P)
{
  exhale acc(x.f, rd);
}
```

At this point we could have decided not to use SIL’s built-in call statement and instead encode synchronous method calls as a series of exhale statements, followed by inhaling the callee’s postcondition. While that would have been equivalent from a verification perspective, we would still be throwing away information: the original program’s call graph.

In order to still use SIL’s call statement, we need to keep track of the “remaining” permissions while constraining  $k_c$  without actually giving away these permissions, otherwise the verification of the call statement would fail. We cannot simply create a copy of the permission mask as a whole and have exhale operate on that instead. SIL at least allows us to look up individual entries of the permission mask via the  $\text{perm}(x.f)$  term. We use that ability to manually create and maintain a permission map of our own.

Like the permission mask in the Boogie-encoding of Chalice, this data structure must map heap locations, represented as pairs of an object reference and a field identifier, to permission amounts. At this time, SIL has no reified field identifiers. So in order to distinguish locations (pair of an object reference and a field), the Chalice2SIL translator assigns a unique integer number to each field in the program.

The only way to populate this map, is to “copy” the current state of the actual permission mask entry by entry via the  $\text{perm}(x.f)$  term. Unfortunately, we can’t do this in one big “initialization” block, since some of the object reference expressions that occur on the right-hand-side of implications might not be defined outside of that implication.

We could expand implications in the precondition twice: once for initializing our permission map, and once to actually simulate the exhales and constraining of  $k_c$ , but there is a more concise way.



Listing 12: Translation sketch for a method call involving fractional read permissions and the precondition `acc(r.val,rd)&& p ==> acc(r.val)`

```

var k_c : Permission
var m : Map[Pair[ref, Integer], Permission];
var m_0 : Map[Pair[ref, Integer], Permission];
assume 0 < k_c && 1000*k_c < k_m;
assume m == m_0;
// acc(r.val,rd)
assume m_0[(r,1)] == perm(r.val);
assert 0 < m[(r,1)];
assume k_c < m[(r,1)];
m[(r,1)] := m[(r,1)] - k_c;
// p ==> acc(r.val,rd)
if(p){
  assume m_0[(r,1)] == perm(r.val);
  assert 0 < m[(r,1)];
  assume k_c < m[(r,1)];
  m[(r,1)] := m[(r,1)] - k_c;
}
// finally, the actual call
call () := m(r,p,k_c);

```

We start out with two fresh map variables  $m$  and  $m_0$ . The former,  $m$ , is the permission map we are going to update while constraining  $k_c$ , whereas  $m_0$  represents the state of the permission map immediately before the method call. We let the SIL verifier assume that the two maps are identical initially and later add more information about  $m$ 's initial state by providing assumptions about  $m_0$ .

After  $k_c$  is sufficiently constrained, we just emit a call to our target method. The SIL verifier will have to exhale the precondition (giving away the permissions it mentions), havoc heap locations that the caller has lost all permissions to, then inhale the postcondition (receiving permissions it mentions) and finally assign results to local variables as necessary.

## 3.2 Asynchronous method calls (Fork-Join)

At this time, SIL only provides synchronous call statements. We therefore have to fall back to just exhaling the precondition on fork and inhaling the postcondition on join. The challenging aspect of verifying asynchronous method calls is establishing the link between a join and the corresponding fork. Old expressions, in particular, are difficult to capture in SIL without a dedicated call statement.

### 3.2.1 Translation of fork

The translation of the Chalice fork statement seems, at least at first, relatively straightforward: exhale the method's precondition and create a token object with a boolean field called "joinable" set to true. But how would we then translate the corresponding join

$k_m$  : read fraction selected for the surrounding scope  
 $e$  : terms  
 $P, Q$  : expressions  
 $p, q$  : permission amount  
 $f$  : field  
 $n$  : integer  
 $g$  : function  
 $G$  : precondition of  $g$   
 $x_1, \dots, x_n$  : parameters of  $g$   
 $m[e]$  : look up map entry with key  $e$  in map  $m$   
 $P[x/e]$  :  $P$ , but with  $e$  substituted for  $x$

Figure 1: Meaning of names used below.

$$\begin{aligned}
H(p * q, h) &= H(p, h) \wedge H(q, h) \\
H(p + q, h) &= H(p, h) \wedge H(q, h) \\
H(p - q, h) &= H(p, h) \wedge H(q, \neg h) \\
H(p * n, h) &= H(p, h) \\
H(k_m, h) &= \neg h \\
H(\_, h) &= \text{false}
\end{aligned}$$

Figure 2: Helper function that determines whether a permission amount expression can be used to constrain  $k_c$ .

$$\begin{aligned}
E\llbracket \text{acc}(e.f, p) \rrbracket_{\text{Ch}} &= \llbracket \text{perm}(E\llbracket e \rrbracket_{\text{Ch}}, f) < E\llbracket p \rrbracket_{\text{Ch}} \rrbracket_{\text{SIL}} \\
E\llbracket e \rrbracket_{\text{Ch}} &\text{ translates expression } e \text{ to SIL terms and expressions} \\
R\llbracket P \rrbracket_{\text{Ch}} &= \llbracket \\
&\quad \text{var } k_c, m, m_0; \\
&\quad \text{inhale } 0 < k_c \wedge k_c * 1000 < k_m; \\
&\quad \text{inhale } m = m_0; \\
&\quad T\llbracket P \rrbracket_{\text{Ch}} \rrbracket_{\text{SIL}} \\
T\llbracket P \wedge Q \rrbracket_{\text{Ch}} &= \llbracket T\llbracket P \rrbracket_{\text{Ch}}; T\llbracket Q \rrbracket_{\text{Ch}} \rrbracket_{\text{SIL}} \\
T\llbracket e \Rightarrow Q \rrbracket_{\text{Ch}} &= \llbracket \text{if } (E\llbracket e \rrbracket_{\text{Ch}}) \{ T\llbracket Q \rrbracket_{\text{Ch}} \} \rrbracket_{\text{SIL}} \\
T\llbracket \text{acc}(x.f, p) \rrbracket_{\text{Ch}} &= \text{if } H(E\llbracket p \rrbracket_{\text{Ch}}, \text{false}) \llbracket \\
&\quad \text{exhale } D\llbracket x \rrbracket_{\text{Ch}}; \\
&\quad \text{inhale } m_0[(E\llbracket x \rrbracket_{\text{Ch}}, f)] = \text{perm}(E\llbracket e \rrbracket_{\text{Ch}}, f); \\
&\quad \text{exhale } 0 < m[(E\llbracket x \rrbracket_{\text{Ch}}, f)]; \\
&\quad \text{inhale } E\llbracket p \rrbracket_{\text{Ch}} < m[(E\llbracket x \rrbracket_{\text{Ch}}, f)]; \\
&\quad m[(E\llbracket x \rrbracket_{\text{Ch}}, f)] := m[(E\llbracket x \rrbracket_{\text{Ch}}, f)] - E\llbracket p \rrbracket_{\text{Ch}} \rrbracket_{\text{SIL}} \\
T\llbracket \text{acc}(x.f, p) \rrbracket_{\text{Ch}} &= \text{otherwise } \llbracket \text{skip} \rrbracket_{\text{SIL}}
\end{aligned}$$

Figure 3: Translation schemes.  $R$  generates code that constrains a fresh  $k_c$  according to the method precondition/loop invariant  $P$ .  $T$  recursively translates  $P$  to constraint  $k_c$ .  $D\llbracket x \rrbracket_{\text{Ch}}$  determines whether  $x$  is well-defined. See figure 4.

$$\begin{aligned}
D[P \Rightarrow Q]_{\text{Ch}} &= \llbracket D[P]_{\text{Ch}} \wedge (E[P]_{\text{Ch}} \Rightarrow D[Q]_{\text{Ch}}) \rrbracket_{\text{SIL}} \\
D[\text{if}(e) \ P \ \text{else} \ Q]_{\text{Ch}} &= \llbracket D[e]_{\text{Ch}} \wedge (E[e]_{\text{Ch}} \Rightarrow D[P]_{\text{Ch}}) \\
&\quad \wedge (\neg E[e]_{\text{Ch}} \Rightarrow D[Q]_{\text{Ch}}) \rrbracket_{\text{SIL}} \\
D[e.f]_{\text{Ch}} &= \llbracket D[e]_{\text{Ch}} \wedge \neg (E[e]_{\text{Ch}} = \text{null}) \\
&\quad \wedge 0 < \text{perm}(E[e]_{\text{Ch}}, f) \rrbracket_{\text{SIL}} \\
D[e.g(a_1, \dots, a_n)]_{\text{Ch}} &= \llbracket D[e]_{\text{Ch}} \wedge \neg (E[e]_{\text{Ch}} = \text{null}) \\
&\quad \wedge E[G[\text{this}/e, x_1/a_1, \dots, x_n/a_n]]_{\text{Ch}} \rrbracket_{\text{SIL}}
\end{aligned}$$

Figure 4: Translation scheme for ensuring the definedness of an expression.

statement(s)? The method’s postcondition is formulated in terms of the method’s return values and parameters. In general we no longer have access to the latter. The join might happen in a different method, but even if it occurs in the same method as the fork, the heap and the values of local variables could have changed in the meantime. Ideally, we could somehow capture the entire program state and store it in or associate it with the token at the fork statement. At the join statement, we would then evaluate (inhale) the method’s postcondition in terms of that program state.

Sadly, SIL currently has no such mechanism. It does have **old** expressions but they are hardwired to refer to the pre-state of the surrounding method (the state immediately prior to a call to that method). Fortunately, we don’t actually need to capture the entire program heap. The set of values that might be missing at the join site are the arguments and the values of old expressions. Since this includes at most all arguments and old expression, we can generate a ghost field on the token to “transport” each of values from the fork site to the join site.

Chalice2SIL generates one ghost field for each method argument and one ghost field for each old expression in the method’s postcondition. Just before the `exhale` statement of a `join`, it assigns the effective arguments to the argument ghost fields of the token. It then evaluates the old expressions of the method’s postcondition and assigns the results to the corresponding ghost fields.

There is just one more complication to take care of: old expressions can appear on the right-hand-side of implications, where they might only be defined part of the time (missing permissions and null references). Unfortunately, just expanding implications into if-statements, like we did when constraining  $k_c$ , is not an option because the left-hand-side of the implication could be a return value, which is of course only available at the join site. Instead, we walk over each old expression and generate a set of conditions that need to be satisfied for the expression to be defined at the fork site.

Need to mention that  $k_c$  is computed in exactly the same way as for synchronous calls?

### 3.2.2 Translation of `join`

With most of the hard work done when the thread was forked, the translation of a `join` statement is relatively straightforward. First, we must assert that the token is still joinable (we also need write-access to that field in order to set it to false). Then we inhale the

Listing 13: Example of Chalice program featuring fork and join of method with a possibly undefined **old** expression.

```
1 class Cell { var f : int; }
2 class SuperCell { var cell : Cell; }
3
4 class Main {
5     method parallel(d : SuperCell) returns (r: bool)
6         requires d != null ==> acc(d.cell, rd) && d.cell != null
7             && acc(d.cell.f, rd) && d.cell.f == 5
8         ensures r == (d != null)
9         ensures r ==> old(d.cell.f == 5)
10        ensures r ==> (acc(d.cell, rd) && acc(d.cell.f, rd))
11    {
12        r := d != null;
13    }
14
15    method main(d : SuperCell, c : Cell)
16        requires acc(d.cell) && acc(c.f)
17        ensures acc(d.cell) && acc(c.f)
18    {
19        var r : bool;
20        d.cell := c;
21        c.f := 5;
22        fork tk := parallel(d)
23        assert c.f == 5; // still have read-access
24        join r := tk;
25        assert r;
26    }
27 }
```

Listing 14: Translation of the fork statement on line 22 in listing 13.

```

var tk : ref;
tk := new ref;
inhale acc(tk.joinable,write);
tk.joinable := true;
// constrain k_c, the read fraction for this call
...
// store arguments in token
inhale acc(tk.this,write);
tk.this := this;
inhale acc(tk.d,write);
tk.d := d;
inhale acc(tk.k_m,write);
tk.k_m := k_c;
//store old values in token
inhale acc(tk.old1,write);
if(d != null && 0 < perm(d.cell) && d.cell != null && 0 < perm(d.cell.f)){
    tk.old1 := (d.cell.f == 5);
}
// "perform" the asynchronous call by exhaling the callee's precondition
exhale this != null && 0 < k_c && k_c < write &&
    d != null ==> acc(d.cell, k_c) && d.cell != null
    && acc(d.cell.f, k_c) && d.cell.f == 5

```

Listing 15: Translation of the join statement on line 24 in listing 13.

```

exhale tk.joinable // SIL verifier also needs to assert that tk != null
inhale acc(tk.r,write) && tk.r == (tk.d != null)
    && tk.r ==> tk.old1
    && tk.r ==> acc(tk.d.cell, tk.k_m) && acc(tk.d.cell.f, tk.k_m);
r := tk.r;
tk.joinable := false;

```

method's postcondition using the ghost fields on the token as substitutions for the arguments and old expressions. Finally, we have to assign the results of the asynchronous computation to the variables indicated by the Chalice programmer.

A detail worth mentioning is the representation of results for the inhale statement. Chalice2SIL also creates ghost fields on the token for results. Since a token is only ever joined once, we can safely inhale the permissions to access those result fields. Conceptually, by joining with the current thread, the forked thread transfers access to its results along with all other permissions from its postcondition.

Alternatively, we could have used fresh local variables to represent result values. The only advantage that ghost fields provide, is that we *don't* need to introduce new variables.

The accessibility of all the other ghost fields on the token requires a bit more work. Naturally, tokens can also be passed to other threads and joined there. The requirement that

$$\begin{aligned}
F[\mathbf{old}(e)]_{\text{Ch}} &= \llbracket \\
&\quad \mathbf{inhale} \text{ acc } (t.f_e, \text{write}) \\
&\quad \mathbf{if} (D[e]_{\text{Ch}}) \{ \\
&\quad \quad t.f_e := E[e]_{\text{Ch}} \\
&\quad \} \\
F[P]_{\text{Ch}} &= \text{descend into } P \text{ to find all } \mathbf{old} \text{ expressions and terms}
\end{aligned}$$

Figure 5: Assign  $\mathbf{old}(e)$  value to its own field  $f_e$ , if  $e$  is defined.

the joining thread has exclusive access to the joinable field ensures that only one thread can join on a given token. Now, while the ghost fields on the token might be invisible to the Chalice programmer, SIL does not distinguish between ghost fields and ordinary fields in any way. We need to make sure that every method that tries to access any of the ghost fields actually has permissions to do so.

Fortunately, ghost fields on a token are only accessed when we also have permission to access the joinable field on that token and it is the Chalice programmer’s burden to ensure that a thread has this permission when attempting to join on a token. If we could somehow link the amount of permission a thread has to each of the ghost fields to the amount of permission it holds to joinable, we would always end up with a sufficient amount of permission for the ghost fields.

While SIL provides no built-in support for linking fields together accessibility-wise, we can achieve a similar effect by translating every accessibility predicate for joinable as an accessibility predicate for that *and* all ghost fields (with the same amount of permission for each). That way, we can be sure that whenever a thread holds full permissions to a joinable field, it also holds full permissions to all ghost fields on the token. More formally, given a token  $t$ , a permission amount  $p$ , ghost fields  $a_1 \cdots a_k$  (the arguments) and  $o_1 \cdots o_n$  (evaluated old expressions), we apply the following transformation:

$$\begin{aligned}
&\llbracket \text{acc } (t.\text{joinable}, p) \rrbracket_{\text{SIL}} \\
&\quad \text{becomes} \\
&\llbracket \text{acc } (t.\text{joinable}, p) \wedge \text{acc } (t.\text{this}, p) \wedge \\
&\quad \wedge \text{acc } (t.a_1, p) \wedge \text{acc } (t.a_2, p) \wedge \cdots \wedge \text{acc } (t.a_k, p) \wedge \\
&\quad \wedge \text{acc } (t.o_1, p) \wedge \text{acc } (t.o_2, p) \cdots \text{acc } (t.o_n) \rrbracket_{\text{SIL}}
\end{aligned}$$

### 3.2.3 Limitations of the current fork-join implementation

Joining a thread seems deceptively simple when done in the same method the thread was originally forked from. This is because the verifier has seen the assignments to the token ghost fields first hand. When a thread is joined in a separate method, however, that context is not available because both Silicon and the Boogie-based implementation verify each method in complete isolation.

For just joining a thread in a separate method, the programmer needs to pass both the token and write access to the token’s joinable field to the method that performs the joining

Listing 16: Limitations with joining in separate methods

```
class Main{
  var f : int;
  method work()
    requires acc(this.f)
    ensures acc(this.f)
  {
  }

  method main()
    requires acc(this.f)
    ensures acc(this.f)
  {
    fork tk := work();
    call client(tk, this);
  }

  method client(tk : token<Main.work>, obj : Main)
    requires acc(tk.joinable) && tk.joinable
    ensures acc(obj.f) // might not hold
  {
    join tk;
  }
}
```

and ensure that the thread has not been joined already. Unfortunately, the postcondition of an asynchronous method call joined this way is next to useless, because the verifier has no information about the context of the method call. Specifically, the verifier doesn't know anything about the receiver or any of the arguments originally passed to the thread. As a consequence, any clause of the postcondition that mentions the `this` pointer or an argument is useless to the verifier.

Listing 16 demonstrates a simple program that fails to verify because the context of the forked thread is lost when the token is transferred to the callee (`client`). The verifier will complain that there might not be enough permission to satisfy `acc(obj.f)`, because it doesn't know that the `this` pointer used to call `work` refers to the same object as `obj`. We would like to tell the verifier more about how our token was created.

```
requires tk.thisPtr == obj //not valid Chalice expression
```

While the previous example is not valid Chalice code, there is a mechanism that can be used to create similar specifications. Listing 17 shows how the `eval` expression can be used to provide the verifier with the information necessary to prove that the method satisfies its postcondition.

An  $\llbracket \text{eval}(r.a, e) \rrbracket_{\text{Ch}}$  expression consists of three parts: the “context”  $c$  (the token in our case), the description of the “eval state”  $a$  and an expression  $e$  to be evaluated in that state. In our case, we specify a “call state” of the form  $\llbracket \text{fork } r.m(a_1, a_2, \dots, a_k) \rrbracket_{\text{Ch}}$ . Here  $r$  denotes the receiver of the asynchronous method call,  $m$  is the name of the method called



Listing 17: `eval` expression in Chalice

```
method client(tk : token<Main.work>, obj : Main)
  requires acc(tk.joinable) && tk.joinable
  requires eval(tk.fork this.work(), this == obj)
  ensures acc(obj.f)
{ join tk; }
```

and  $a_i$  stand for the arguments originally passed to the method.

Chalice2SIL supports a very limited form of the `eval` expression which covers exactly the use-case outlined above. As long as the `eval` expression binds to a `fork` token and has `true` as its second operand  $e$ , Chalice2SIL translates it as follows:

$$\begin{aligned} V[\llbracket \text{eval}(t.\text{fork } r.m(a_1, a_2, \dots, a_k), \text{true}) \rrbracket]_{\text{Ch}} &= \llbracket E[t]_{\text{Ch}} \neq \text{null} \\ &\quad \wedge E[t]_{\text{Ch}}.\text{this} = E[r]_{\text{Ch}} \\ &\quad \wedge E[t]_{\text{Ch}}.a_1 = E[a_1]_{\text{Ch}} \\ &\quad \vdots \\ &\quad \wedge E[t]_{\text{Ch}}.a_k = E[a_k]_{\text{Ch}} \rrbracket_{\text{SIL}} \end{aligned}$$

This small extension is just expressive enough to associate tokens with parts of the context they were forked from, allowing the joining method to actually take advantage of the postcondition of the forked method.

### 3.3 Predicates and Functions

Predicates and functions were part of SIL since its inception, in a form that very much resembles the functions and predicates from Chalice. As a result, the translation of functions and predicates from Chalice to SIL is relatively straightforward.

#### 3.3.1 Predicates

In contrast to abstract fractional read permissions in methods (`rd`), which can assume a different value for each invocation, the fraction used in predicates remains fixed. This is essential to ensure that the predicate holds the same amount of permission regardless of where it was folded. Otherwise the user would have to specify exactly how much permission a predicate contains, which rather defeats the purpose of predicates (information hiding).

Abstract read permissions mentioned in predicates are thus fixed. To implement this, we define an uninterpreted constant function `globalPredicateReadFraction()` and declare that this global read fraction is strictly positive and less than full/write permission.

It is possible that, in the future, we'd like to have different fractions for different predicates or even different fractions for each combination of object (this-pointer) and predicate. To support this scenario, we don't insert references to `globalPredicateReadFraction()`

directly into the generated SIL program and instead use an intermediate function `predicateReadFraction` (`int`, `ref`), also uninterpreted. Analogously to the “field identifiers” that we used to index into our copy of the permission mask when we constrain the read permission fraction for method calls, we generate unique “predicate identifiers” to distinguish between different predicates on the same object. However, at the moment every SIL program that Chalice2SIL generates also contains an axiom that makes predicate read fractions effectively global:

$$\forall i, r. \text{predicateReadFraction}(i, r) = \text{globalPredicateReadFraction}()$$

This constraint makes it easier to transfer abstract read permissions between predicates, giving `acc(x.f, rd)` a fixed meaning across all predicates. Outside of predicates, the user can refer to the same amount of permission by explicitly mentioning the predicate in an argument to the abstract read permission. Given two object references  $x$  and  $y$ , a field  $f$ , a predicate  $p$  and a corresponding predicate identifier  $i_p$ , Chalice2SIL translates an accessibility predicate that involves an abstract read fractions for predicates as follows:

$$\llbracket \text{acc}(x.f, \text{rd}(y.p)) \rrbracket_{\text{Ch}} = \llbracket \text{acc}(E\llbracket x.f \rrbracket_{\text{Ch}}, \text{predicateReadFraction}(i_p, E\llbracket y \rrbracket_{\text{Ch}})) \rrbracket_{\text{SIL}}$$

### 3.3.2 Functions

Ideally, we would want to encode abstract read permissions in functions the same way we encode them when they occur in method pre/postconditions: Constraining a different fraction for each call site. Unfortunately, the technique we used for read fractions in method calls would not work for function calls, because our solution is based on “mutating” a data structure in an imperative fashion. Function calls can occur in method pre/postconditions and even inside other function bodies. All of those contexts are pure, and thereby inherently incompatible with how abstract read permission fractions are fixed for method calls.

But functions being pure also gives us more freedom since we don’t have to make the distinction between read permissions and read-write permissions (functions aren’t allowed to modify the heap anyway). Similarly, as functions always automatically return the same amount of permission that they received, it doesn’t really matter exactly “how much” permission a function has, only to which heap locations it has access. As a result, Chalice2SIL translates permissions in function preconditions as `rd*`, which means “any read-permission”.

$$\llbracket \text{acc}(x.f, \text{rd}*) \rrbracket_{\text{Ch}} = \llbracket \exists a, 0 < a < \text{write}. \text{acc}(x.f, a) \rrbracket_{\text{SIL}}$$

### 3.4 Monitors with Deadlock Avoidance

As with predicates, the permissions stored in a monitor need to be fixed and cannot be chosen every time we lock over an object. Otherwise we’d have to track what amount of permission the monitor holds onto at any point in the program. So, as with predicates, we define an uninterpreted function that represents the fraction of permission that the verifier uses whenever it encounters an abstract fractional read permission in a monitor. To make mixing monitors and predicates easier, we use the same global fraction for monitors as we used for predicates.

The really difficult part, however, is the implementation of the locking and deadlock prevention itself. In this section we present our partial solution in the hope that it will help in finding a working implementation of deadlock avoidance in SIL, or failing that, in the hope that it at least serves as a demonstration of the limits of SIL.

### 3.4.1 Deadlock Prevention and Locking

For locking and deadlock prevention we need to add two kinds of information to each object: a boolean indicating whether the object in question is locked and a *mu* value that indicates the object's position in the locking order. To recapitulate, mu values are part of the partially ordered set  $(Mu, \ll)$ . Only one concrete value exists: *lockbottom*, the single smallest element of Mu. All other values of Mu are kept abstract and only described in terms of their  $\ll$ -relation to one another.

The key idea behind deadlock prevention in Chalice, that a thread can only acquire a lock on monitor/object *m* if it is clear that that monitor/object is higher in the locking order than any other monitor that that thread already has a lock on. In more concrete terms, locking is allowed when  $\forall o \in \text{objects}. o = m \vee \neg \text{holds}(o) \vee o.\text{mu} \ll m.\text{mu}$ .

To use this expression in a SIL assertion (an exhale statement prior to acquiring a lock), we need to make sure that the relevant locations are well defined. To ensure that we can access *m.mu*, we first check  $m \neq \text{null} \wedge 0 < \text{perm}(m.\text{mu})$ . But then we run into the problem that we cannot possibly require access to the mu-fields on all objects *o*. One could try guard references to *o.mu* with an implication:

$$0 < \text{perm}(o.\text{mu}) \Rightarrow \neg \text{holds}(o) \vee o.\text{mu} \ll m.\text{mu}$$

Unfortunately that is not correct. While a method needs read-access to the mu field to acquire the lock on an object *o*<sub>1</sub>, it can then give away all permissions to fields of *o*<sub>1</sub> (it could fork a thread and hand complete control over that object to the forked thread), while still holding the lock on the monitor associated with *o*<sub>1</sub>. With the assertion above, temporarily not having any permission to access to *o*<sub>1</sub>.mu would allow us to acquire the lock on a monitor that is not higher in the locking order than *o*<sub>1</sub>.

We need to associate each object with a mu and a flag that indicates whether the object's monitor is acquire by the current thread. The only mechanism that SIL provides for associating information with an object are fields but using fields is not an option. The Boogie-based implementation uses additional masks and that is what we will try to emulate in our approach.

We add another hidden parameter to each method, the `\$CurrentThread: ref`. This object has two fields `heldMap` and `muMap`, mapping from references to boolean values and from references to mu values, respectively. Via their preconditions, each method gets to assume the following:

$$\begin{aligned} & \text{currentThread} \neq \text{null} \wedge \text{acc}(\text{currentThread.heldMap}, \text{write}) \\ & \wedge \text{acc}(\text{currentThread.muMap}, \text{write}) \end{aligned}$$

However, the mu field is not an implementation detail of Chalice2SIL or the Boogie-based Chalice verifier, but an actual field that the user of Chalice has to take into consideration when dealing with monitors. The greatest challenge is thus to keep what the verifier knows about the muMap and what it knows about mu fields in synchrony. To that end, every

occurrence of an accessibility predicate that mentions `mu` will be translated according to the following rule:

$$\llbracket \text{acc}(x.\text{mu}, a) \rrbracket_{\text{Ch}} = \llbracket \text{acc}(x.\text{mu}, a) \wedge \text{currentThread.muMap}[x] = x.\text{mu} \rrbracket_{\text{SIL}}$$

This makes sure that whenever we gain access to an `mu` field, we also get a matching entry in the `currentThread`'s `muMap`. Or, conversely, whenever we give away permission to a `mu` field, we must also ensure that the `muMap` is up to date.

newobj

share

acquire

release

unshare

In the postcondition of each method, we make sure that the `mu` and `help` maps are in a consistent state. The programmer uses the **lockchange** declaration to list all objects it changed the lock state of. Without any **lockchange** declarations, all method postcondition contain the following assertions:

```
acc(currentThread.heldMap, write) ∧ acc(currentThread.muMap, write)
∧ ∀ oc ∈ objects. old(currentThread.heldMap)[oc] = currentThread.heldMap[oc]
∧ ∀ od ∈ objects. (currentThread.heldMap[od]) ⇒
    old(currentThread.muMap[od]) = currentThread.muMap[od]
```

The second line demands that the locking state of any object must not have been changed. The last two lines ensure that the lock level of anything that is currently locked doesn't change. When the user adds **lockchange** declarations, the bodies of the last two quantifiers are guarded by an implication similar to:

$o \notin \text{lockchange} \Rightarrow \text{body} \dots$

## 4. Evaluation

### 4.1 Chalice2SIL+Silicon versus Syxc

Compare Chalice2SIL+Silicon with Syxc (how complete is Chalice2SIL)

### 4.2 SIL as a translation target/verification intermediate language

- Term/Expression distinction (technical)
- P-expressions/terms versus non-P-expressions/terms (technical)
- control flow (need for a while loop construct)
- Some way of capturing state (more general old expression, "transaction"-like scopes)

## 5. Conclusion

### The conclusion

## A Full SIL Term and Expression Grammar

```
 $\langle Expr \rangle ::= 'acc' (\langle Location \rangle, \langle Term \rangle)$   
|  $'old' (\langle Expr \rangle)$   
|  $'unfolding' \langle Term \rangle . \langle pred-id \rangle 'by' \langle Term \rangle 'in' \langle Expr \rangle$   
|  $\langle Term \rangle == \langle Term \rangle$   
|  $\langle unary-op \rangle \langle Expr \rangle$   
|  $\langle binary-op \rangle \langle Expr \rangle$   
|  $\langle dom-pred-id \rangle (\{ \langle Term \rangle , \dots \})$   
|  $\forall \langle logical-var-id \rangle : \langle DataType \rangle :: (\langle Expr \rangle)$   
|  $\exists \langle logical-var-id \rangle : \langle DataType \rangle :: (\langle Expr \rangle)$   
|  $\langle GExpr \rangle$   
  
 $\langle Location \rangle ::= \langle Term \rangle . \langle field-id \rangle$   
|  $\langle Term \rangle . \langle pred-id \rangle$   
  
 $\langle PExpr \rangle ::= 'acc' (\langle PLocation \rangle, \langle PTerm \rangle)$   
|  $'unfolding' \langle PTerm \rangle . \langle pred-id \rangle 'by' \langle PTerm \rangle 'in' \langle PExpr \rangle$   
|  $\langle PTerm \rangle == \langle PTerm \rangle$   
|  $\langle unary-op \rangle \langle PExpr \rangle$   
|  $\langle binary-op \rangle \langle PExpr \rangle$   
|  $\langle dom-pred-id \rangle (\{ \langle PTerm \rangle , \dots \})$   
|  $\langle GExpr \rangle$   
  
 $\langle PLocation \rangle ::= \langle PTerm \rangle . \langle field-id \rangle$   
|  $\langle PTerm \rangle . \langle pred-id \rangle$   
  
 $\langle DExpr \rangle ::= \langle DTerm \rangle == \langle DTerm \rangle$   
|  $\langle unary-op \rangle \langle DExpr \rangle$   
|  $\langle binary-op \rangle \langle DExpr \rangle$   
|  $\langle dom-pred-id \rangle (\{ \langle DTerm \rangle , \dots \})$   
|  $\forall \langle logical-var-id \rangle : \langle DataType \rangle :: (\langle DExpr \rangle)$   
|  $\exists \langle logical-var-id \rangle : \langle DataType \rangle :: (\langle DExpr \rangle)$   
|  $\langle GExpr \rangle$   
  
 $\langle GExpr \rangle ::= \langle PExpr \rangle == \langle PExpr \rangle$   
|  $\langle UnaryOp \rangle \langle PExpr \rangle$   
|  $\langle BinaryOp \rangle \langle PExpr \rangle$   
|  $\langle dom-pred-id \rangle (\{ \langle PExpr \rangle , \dots \})$   
|  $'True'$   
|  $'False'$   
  
 $\langle UnaryOp \rangle ::= '\neg'$   
  
 $\langle BinaryOp \rangle ::= '\wedge' \mid '\vee' \mid '\equiv' \mid '\Rightarrow'$ 
```

```

<Term> ::= 'if' <Term> 'then' <Term> 'else' <Term>
| 'old' ( <Term> )
| <func-id> ( { <Term> , ... } )
| <dom-func-id> ( { <Term> , ... } )
| 'unfolding' <Term> . <pred-id> 'by' <Term> 'in' <Term>
| ( <Term> ) : <DataType>
| <Term> . <field-id>
| 'perm' ( <Location> )
| 'write'
| '0'
| 'E'
| <GTerm>

<PTerm> ::= 'if' <PTerm> 'then' <PTerm> 'else' <PTerm>
| <var-id>
| <func-id> ( { <PTerm> , ... } )
| <dom-func-id> ( { <PTerm> , ... } )
| 'unfolding' <PTerm> . <pred-id> 'by' <PTerm> 'in' <PTerm>
| ( <PTerm> ) : <DataType>
| <PTerm> . <field-id>
| <GTerm>

<DTerm> ::= 'if' <DTerm> 'then' <DTerm> 'else' <DTerm>
| <logical-var-id>
| <dom-func-id> ( { <DTerm> , ... } )
| <GTerm>

<GTerm> ::= 'if' <GTerm> 'then' <GTerm> 'else' <GTerm>
| <integer-literal>
|
| <dom-func-id> ( { <GTerm> , ... } )

```

## References

- [BDJ<sup>+</sup>06] M. Barnett, R. DeLine, B. Jacobs, B.-Y. E. Chang, and K. R. M. Leino. Boogie: A modular reusable verifier for object-oriented programs. In *Formal Methods for Components and Objects: 4th International Symposium, volume 4111 of Lecture Notes in Computer Science*. Springer, 2006.
- [Boy03] J. Boyland. Checking interference with fractional permissions. In Radhia Cousot, editor, *Static Analysis, volume 2694 of Lecture Notes in Computer Science*, pages 55–72. Springer Berlin / Heidelberg, 2003. 10.1007/3-540-44898-5\_4.
- [dMB08] L. de Moura and N. Bjørner. Z3: An efficient smt solver. In *In TACAS 2008, volume 4963 of LNCS*, pages 337–340. Springer, 2008.
- [HLMS11] S. Heule, K. R. M. Leino, P. Müller, and A. J. Summers. Fractional permissions without the fractions. *Formal Techniques for Java-like Programs (FTfJP)*, 2011.